Modeling Sectoral Labor Transitions with WiNDC

Presenter: Jonathon Becker (NREL) -- *Jon.Becker@NREL.gov*

Collaborator: Jared Carbone (Colorado School of Mines)

Overview

<u>"Modeling age-cohort employment responses to a new environmental regulation"</u>

Jared Carbone and Jonathon Becker

Past work funded by the Environmental Protection Agency

- How do employment responses to new environmental regulations differ across age cohorts and sectors?
- One (of many) modeling challenges:
  - How to capture possibilities for labor force transitions by demographic group?
- Wisconsin National Data Consortium (WiNDC) has been developing features for distributional analysis
  - a key component of distributional analysis is how a household's jobs are impacted

Today's Discussion

- Determining benchmark transition rates
- CGE model calibration
- An example calibration using WiNDC blueNOTE
- A test counterfactual case:
  - 30% coal input tax
- Link to example code on github: https://github.com/jonmbecker/ltran

Determining Benchmark Transition Rates

Data: IPUMS Basic Monthly Census Population Survey (CPS)

- Tracks individuals:
  - 4 on -- 8 off -- 4 on
  - Maximum of 8 survey responses in a 16 month period
- Data points
  - Labor force status, employment status, age, industry, occupation, etc
- Transition flows estimated with a proportional estimation by tracking first and last observed industry by person

Provides benchmark shares of total population attributable to inter- and intra-industry worker flows

Example Transition Rates from CPS data

Population weighted:
Share of total population making the transition from origin sector (o) to destination sector (s)

Population weighted transition rate from origin (o) to destination (s) (%)

		Destination (s)							
		FFL	ELE	EIT	TRN	MFG	SRV	AOG	Sum(s,*)
	FFL	0.33	0.01	0	0.01	0.01	0.02	0.07	0.45
	ELE	0	0.18	0	0	0	0.01	0.04	0.23
	EIT	0.01	0	0.75	0.01	0.09	0.06	0.19	1.11
Origin (o)	TRN	0.01	0	0.01	1.83	0.05	0.2	0.44	2.54
	MFG	0.01	0.01	0.1	0.05	4.1	0.4	0.96	5.63
	SRV	0.02	0.02	0.07	0.21	0.42	28.19	4.94	33.87
	AOG	0.08	0.04	0.2	0.51	1.03	5.73	48.58	56.17
									100

Benchmark transition rates:
Share of workers with initial experience in sector "o" who transition to sector "s"

Obtained by dividing each element of the population weighted matrix by the sum over "s"

Benchmark transition rate from origin (o) to destination (s) (%)

		Destination (s)							
		FFL	ELE	EIT	TRN	MFG	SRV	AOG	Sum(s,*)
	FFL	74.36	1.3	1.07	1.5	2.14	4.52	15.11	100
	ELE	1.61	74.54	0.36	0.65	1.88	6.23	14.73	100
	EIT	0.46	0.06	67.38	0.93	8.29	5.68	17.2	100
Origin (o)	TRN	0.35	0.08	0.42	72.12	2.1	7.8	17.14	100
	MFG	0.21	0.09	1.73	0.94	72.78	7.17	17.08	100
	SRV	0.07	0.06	0.2	0.62	1.24	83.24	14.57	100
	AOG	0.14	0.08	0.35	0.9	1.83	10.21	86.49	100

Calibrating benchmark labor flows in CGE

Constant Elasticity of Transformation (CET) function:

$$LS_{r,o} = A\left(\sum_s \theta^{LT}_{r,o,s} * LD^{\rho^{LT}_{r,o}}_{r,o,s}\right)^{1/\rho^{LT}_{r,o}} \quad , \quad \sigma^{LT}_{r,o} = 1/(\rho^{LT}_{r,o} - 1)$$

- $\theta^{LT}_{r,o,s}$ benchmark rate at which workers of type o transition to sector s
- $\sigma^{LT}_{r,o}$ elasticity of transformation across sectors

Calibrating benchmark labor flows in CGE

Solve system of linear equations $\mathbf{b} = \mathbf{Ax}$

SOLVE for benchmark supply of labor ($VFO_{r,o}$), GIVEN benchmark labor demand ($ld0_{r,s}$) and labor transition rates ($ltran_{r,o,s}$):

$$ld0_{r,s} = \sum_{o} \left[VFO_{r,o} * ltran_{r,o,s} \right]$$

such that demand for labor in region r of experience type o in sector s is:

$$ld0trn_{r,o,s} = VFO_{r,o} * ltran_{r,o,s}$$

where $ld0trn_{r,o,s}$ enters the model.

WiNDC Toy: Create placeholder transition data

```
* set o used for origin sector
alias(s,o,oo);

parameter lshr(r,o,s)        labor transition rate from origin (o) to destination sector (s);
parameter numberl(r,o,s)     randomization of transition rates;
parameter popr(r,o)          household disagg share by o;

* Random number for toy transition rates
numberl(r,o,s)$ld0(r,s) = uniform(0.9,1.1);
* Rescale according to the size of sectoral labor demand
numberl(r,o,s)$(ld0(r,s)) = numberl(r,o,s)*ld0(r,s)/sum((s.local),ld0(r,s));

* Assume 75% same sector and distribute non same sector across remaining 25%
loop((r,o),
lshr(r,o,s)$(not sameas(o,s)) =
    0.25*numberl(r,o,s)$(not sameas(o,s))/sum(s.local,numberl(r,o,s)$(not sameas(o,s)));
);

lshr(r,o,s)$(sameas(o,s)) = 0.75;
```

		destination (s)										
census		con	trn	oil	col	ele	eint	omnf	osrv	roe	gas	cru
ESC	con	75.00%	0.76%	0.13%	0.03%	0.19%	4.91%	0.94%	11.61%	6.41%	0.01%	0.01%
	trn	1.38%	75.00%	0.12%	0.03%	0.19%	5.04%	0.96%	11.21%	6.05%	0.01%	0.02%
	oil	1.20%	0.69%	75.00%	0.03%	0.15%	4.72%	0.80%	11.22%	6.15%	0.01%	0.01%
	col	1.43%	0.70%	0.12%	75.00%	0.18%	4.43%	0.90%	10.83%	6.38%	0.01%	0.02%
origin (o)	ele	1.26%	0.71%	0.11%	0.03%	75.00%	4.36%	0.91%	11.37%	6.22%	0.01%	0.01%
	eint	1.55%	0.88%	0.17%	0.04%	0.20%	75.00%	1.12%	12.93%	8.07%	0.01%	0.02%
	omnf	1.26%	0.64%	0.12%	0.03%	0.17%	4.88%	75.00%	11.86%	6.01%	0.01%	0.02%
	osrv	2.51%	1.27%	0.22%	0.05%	0.29%	7.41%	1.50%	75.00%	11.71%	0.02%	0.03%
	roe	1.61%	0.88%	0.15%	0.04%	0.23%	6.02%	1.00%	15.04%	75.00%	0.01%	0.02%
	gas	1.39%	0.68%	0.11%	0.03%	0.17%	4.78%	0.84%	10.85%	6.13%	75.00%	0.01%
	cru	1.38%	0.71%	0.13%	0.03%	0.17%	4.99%	0.86%	10.06%	6.65%	0.01%	75.00%

WiNDC Toy: Calibrate benchmark labor demand

```
* Solve for benchmark labor supply in origin sector o using transition rate and labor demand
variables
    VFO(r,o)     labor supply in origin sector o
    MN           dummy
;

equations
    main(r,s)
    dummy
;

main(r,s)..      ld0(r,s) =e= sum(o, VFO(r,o)*lshr(r,o,s));
dummy..          MN =e= 0;

model getvfo     /main, dummy/;

MN.l = 0;
VFO.lo(r,o) = 0;

solve getvfo minimizing MN using lp;

* used to disaggregate househols
parameter   oshr(r,o)   share of origin sector o in region r;
oshr(r,o) = VFO.l(r,o)/sum((o.local),VFO.l(r,o));
popr(r,o) = oshr(r,o);

* Origin specific labor demand enters model to inform supply and demand
parameter ld0_o(r,o,s)  origin specific labor demand;
ld0_o(r,o,s) = VFO.l(r,o)*lshr(r,o,s);
```

WiNDC Toy: Create placeholder household disaggregation

```
* Preprocessing for household disaggregation by o
parameter bmk_trev(r)   benchmark tax revenues;

bmk_trev(r) = sum((s,g)$y_(r,s), ty0(r,s) * ys0(r,s,g))
    + sum((g)$a_(r,g),   ta0(r,g)*a0(r,g) + tm0(r,g)*m0(r,g));
*   + taxrevL(r) + sum(s,tk0(r)*kd0(r,s));

parameter tpo(r,o)  benchmark transfers;

tpo(r,o) = (c0(r) - (
    sum(g,yh0(r,g))
    + bopdef0(r)
    + hhadj(r)
    - sum(g,i0(r,g))
    + sum(s,ld0(r,s))
    + sum(s,kd0(r,s))
    ))*popr(r,o)
;

parameter gov0  government deficit;
gov0 = sum(r,sum(g,g0(r,g)) + sum(o,tpo(r,o)) - bmk_trev(r));
```

WiNDC Toy: Modify the WiNDC model

```
* New/updated model variables
$sectors:
    C(r,o)$popr(r,o)           ! Aggregate final demand
    LS(r,o)$popr(r,o)          ! Labor Supply

$commodities:
    PC(r,o)$popr(r,o)          ! Consumer price index
    PL(r,o,s)$ld0_o(r,o,s)     ! Wage
    PELL(r,o)$popr(r,o)        ! Opportunity cost of labor

$consumer:
    RA(r,o)$popr(r,o)          ! Representative agent
    GOVT                       ! Government

$auxiliary:
    TRANS                      ! Transfers
```

WiNDC Toy: Modify the WiNDC model

```
$prod:Y(r,s)$y_(r,s)   s:0 va:1
    o:PY(r,g)          q:(ys0(r,s,g)*(1-tfp(r,s)))       a:GOVT   t:ty(r,s)    p:(1-ty0(r,s))
    i:PA(r,g)          q:(id0(r,g,s)*(1-aeeir(r,g)))     a:GOVT   t:te(r,g,s)
    i:PK(r,s)          q:kd0(r,s)          va:
    i:PL(r,o,s)        q:ld0_o(r,o,s)      va:

$prod:LS(r,o)$popr(r,o)  t:1
    o:PL(r,o,s)        q:ld0_o(r,o,s)
    i:PELL(r,o)        q:(sum(s,ld0_o(r,o,s)))

$prod:C(r,o)$popr(r,o)   s:1
    o:PC(r,o)          q:(c0(r)*popr(r,o))
    i:PA(r,g)          q:(cd0(r,g)*popr(r,o))

$demand:RA(r,o)$popr(r,o)
    d:PC(r,o)          q:(c0(r)*popr(r,o))
    e:PY(r,g)          q:(yh0(r,g)*popr(r,o))
    e:PFX              q:((bopdef0(r) + hhadj(r))*popr(r,o))
    e:PA(r,g)          q:(-(i0(r,g))*popr(r,o))
    e:PELL(r,o)        q:((sum((s,o.local),ld0_o(r,o,s)))*popr(r,o))
    e:PK(r,s)          q:(kd0(r,s)*popr(r,o))
    e:PFX              q:tpo(r,o)   r:TRANS
```

30% coal input tax

% Differences:

The coal input tax leads workers with experience in the coal and electricity sectors to transition away towards other industries

Workers who were transitioning towards coal and electricity, now transition less

Raw Differences:

Coal and electricity workers transition into other industries

Service (osrv) sector workers who were transitioning to the electricity and coal sectors are more likely to stay put

Labor Transitions -- 30% coal good input tax -- % difference from BMK labor flows

origin (o) \ destination (s)	con	trn	oil	col	ele	eint	omnf	osrv	roe	gas	cru
con	0.00	0.00	0.01	-1.85	-2.69	0.00	0.02	0.02	-0.01	-0.04	0.00
trn	0.00	0.00	0.01	-1.85	-2.56	0.00	0.02	0.02	0.00	-0.03	0.01
oil	0.00	-0.01	0.00	-1.88	-2.87	-0.01	0.01	0.01	-0.01	-0.04	0.00
col	1.41	1.41	1.44	-0.47	-2.18	1.44	1.42	1.42	1.39	1.32	1.38
ele	2.02	1.97	2.16	0.77	-0.67	2.05	1.68	1.94	2.08	2.36	2.22
eint	0.00	0.00	0.02	-1.92	-2.72	0.00	0.02	0.02	0.00	-0.03	0.01
omnf	-0.01	-0.01	0.00	-1.84	-2.16	-0.01	0.01	0.00	-0.02	-0.05	0.00
osrv	-0.01	-0.01	0.00	-1.87	-2.25	-0.01	0.01	0.01	-0.01	-0.04	0.00
roe	0.01	0.00	0.02	-1.82	-2.84	0.00	0.02	0.02	0.00	-0.03	0.01
gas	0.04	0.03	0.04	-1.76	-3.16	0.03	0.05	0.05	0.03	-0.01	0.04
cru	0.01	-0.01	0.01	-1.80	-2.90	0.00	0.01	0.01	-0.01	-0.04	0.00

Labor transitions -- 30% coal good input tax -- raw difference from BMK labor flows

origin (o) \ destination (s)	con	trn	oil	col	ele	eint	omnf	osrv	roe	gas	cru
con	0.0028	-0.0001	0.0000	-0.0013	-0.0127	0.0015	0.0009	0.0097	-0.0011	-0.0001	0.0000
trn	0.0001	-0.0003	0.0000	-0.0004	-0.0038	0.0006	0.0003	0.0036	-0.0002	0.0000	0.0000
oil	0.0000	0.0000	0.0006	-0.0001	-0.0009	0.0000	0.0000	0.0004	-0.0001	0.0000	0.0000
col	0.0009	0.0003	0.0001	-0.0142	-0.0001	0.0017	0.0006	0.0074	0.0032	0.0000	0.0000
ele	0.0084	0.0031	0.0007	0.0000	-0.1423	0.0195	0.0053	0.0718	0.0304	0.0002	0.0003
eint	-0.0008	-0.0006	0.0000	-0.0027	-0.0345	0.0282	0.0012	0.0140	-0.0054	-0.0001	0.0000
omnf	-0.0007	-0.0003	0.0000	-0.0008	-0.0074	-0.0002	0.0104	0.0020	-0.0032	-0.0001	0.0000
osrv	-0.0141	-0.0073	0.0002	-0.0294	-0.2865	0.0147	0.0104	0.3885	-0.0806	-0.0012	0.0000
roe	0.0017	0.0001	0.0003	-0.0060	-0.0594	0.0085	0.0050	0.0572	-0.0085	-0.0002	0.0001
gas	0.0000	0.0000	0.0000	0.0000	-0.0003	0.0001	0.0001	0.0004	0.0001	-0.0004	0.0000
cru	0.0000	0.0000	0.0000	0.0000	-0.0004	0.0000	0.0000	0.0002	0.0000	0.0000	0.0003

Concluding Remarks / Caveats

- Labor transitions can be modeled in a static CGE model, such as WiNDC
- Extra model dimensions make calibration trickier
- Quality data is hard to get:
  - CPS data works but comes with some caveats in terms of sparseness
  - Disaggregating coal and gas extraction transitions from CPS an open issue
- A justifiable calibration relies on transformation elasticity estimates
  - A discrete choice model is being tested for determining this elasticity and transition rates (Jared Carbone)
- It could be difficult to justify the household disaggregation under some circumstances
- Unemployment and not in labor force transitions present a complication

References

Rutherford, T.F. and A. Schreiber (2019). "Tools for Open Source, Subnational CGE Modeling with an Illustrative Analysis of Carbon Leakage", Journal of Global Economic Analysis, Volume 4 (2019), No. 2, pp. 1-66. DOI: http://dx.doi.org/10.21642/JGEA.040201AF

Marten, A., Schreiber, A., and Wolverton, A. 2021. SAGE Model Documentation (2.0.1). U.S. Environmental Protection Agency: https://www.epa.gov/environmental-economics/cge-modeling-regulatory-analysis.

Thank you!

www.nrel.gov

Email: Jon.Becker@NREL.gov

NREL/PR-6A20-82781

Appendix

Labor transition